```
In[49]:= <<DiscreteMath`Combinatorica`

In[50]:= Off[General::"spell1"]
```

Problem 19

```
In[51]:= Clear[wa, n];

         wa[0] = 1; wa[1] = 2; wa[2] = 4; wa[3] = 8; wa[4] = 16;

         wa[n_] := wa[n] = 2 * wa[n - 1] + wa[n - 5]

In[54]:= Table[wa[k], {k, 1, 10}]

Out[54]= {2, 4, 8, 16, 33, 68, 140, 288, 592, 1217}
```

Problem 22

```
In[55]:= {{1, 3}, {1, 2, 3}} (* n = 3 ; 2^(3-2) *)

Out[55]= {{1, 3}, {1, 2, 3}}

In[56]:= {{1, 4}, {1, 2, 4}, {1, 3, 4}, {1, 2, 3, 4}} (* n = 4 ; 2^(4-2) *)

Out[56]= {{1, 4}, {1, 2, 4}, {1, 3, 4}, {1, 2, 3, 4}}

In[57]:= {{1, 5}, {1, 2, 5}, {1, 3, 5}, {1, 2, 3, 5},
          {1, 4, 5}, {1, 2, 4, 5}, {1, 3, 4, 5}, {1, 2, 3, 4, 5}}

Out[57]= {{1, 5}, {1, 2, 5}, {1, 3, 5}, {1, 2, 3, 5},
          {1, 4, 5}, {1, 2, 4, 5}, {1, 3, 4, 5}, {1, 2, 3, 4, 5}}

In[58]:= {{1, 6}, {1, 2, 6}, {1, 3, 6}, {1, 2, 3, 6}, {1, 4, 6}, {1, 2, 4, 6},
          {1, 3, 4, 6}, {1, 2, 3, 4, 6}, {1, 5, 6}, {1, 2, 5, 6}, {1, 3, 5, 6},
          {1, 2, 3, 5, 6}, {1, 4, 5, 6}, {1, 2, 4, 5, 6}, {1, 3, 4, 5, 6}, {1, 2, 3, 4, 5, 6}}

Out[58]= {{1, 6}, {1, 2, 6}, {1, 3, 6}, {1, 2, 3, 6}, {1, 4, 6}, {1, 2, 4, 6},
          {1, 3, 4, 6}, {1, 2, 3, 4, 6}, {1, 5, 6}, {1, 2, 5, 6}, {1, 3, 5, 6},
          {1, 2, 3, 5, 6}, {1, 4, 5, 6}, {1, 2, 4, 5, 6}, {1, 3, 4, 5, 6}, {1, 2, 3, 4, 5, 6}}
```

This can be explained by bit strings of length $n-2$. The positions belong to the numbers 2, ..., $n-1$. 1-s will tell you which numbers to include in the sequence.

```
In[59]:= Clear[ns, n];

         ns[2] = 1; ns[3] = 2; ns[n_] := ns[n] = 2 ns[n - 1]

In[61]:= Table[ns[k], {k, 2, 10}]

Out[61]= {1, 2, 4, 8, 16, 32, 64, 128, 256}
```

Problem 23

The problem is to find a recurrence realtion for the number of bit strings of length n which contain at least one occurence of the string 00.

Mathematica can select all such bit strings.

```
In[62]:= Select[Strings[{0, 1}, 3], MemberQ[Partition[#, 2, 1], {0, 0}] &]

Out[62]= {{0, 0, 0}, {0, 0, 1}, {1, 0, 0}}

In[63]:= Select[Strings[{0, 1}, 4], MemberQ[Partition[#, 2, 1], {0, 0}] &]

Out[63]= {{0, 0, 0, 0}, {0, 0, 0, 1}, {0, 0, 1, 0}, {0, 0, 1, 1},
          {0, 1, 0, 0}, {1, 0, 0, 0}, {1, 0, 0, 1}, {1, 1, 0, 0}}

In[64]:= Length[Select[Strings[{0, 1}, 4], MemberQ[Partition[#, 2, 1], {0, 0}] &]]

Out[64]= 8

In[65]:= Table[
          Length[Select[Strings[{0, 1}, k], MemberQ[Partition[#, 2, 1], {0, 0}] &]], {k, 1, 15}]

Out[65]= {0, 1, 3, 8, 19, 43, 94, 201, 423, 880, 1815, 3719, 7582, 15397, 31171}
```

Let $n \geq 3$. Denote by S the set of all bit strings of length n which contain at least one occurence of the string 00. The set S is a disjoint union of three sets: the set S_{00}, the set S_{10}, and the set S_1. The set S_{00} is the set of all bit strings in S which end with 00, The set S_{10} is the set of all bit strings in S which end with 10 and the set S_1 is the set of all bit strings in S which end with 1. The cardinality of S_{00} is 2^{n-2}, the cardinality of S_{10} is a_{n-2}, the cardinality of S_1 is a_{n-1}.

The corresponding recursion is

```
In[66]:= Clear[rs23]; rs23[0] = 0; rs23[1] = 0;
         rs23[n_] := rs23[n] = 2^(n-2) + rs23[n - 1] + rs23[n - 2]
```

The logic here is that 2^{n-2} strings end with 00, rs[n-2] strings end with 10 and rs[n-1] strings end with 1.

```
In[68]:= Table[rs23[k], {k, 0, 18}]

Out[68]= {0, 0, 1, 3, 8, 19, 43, 94, 201, 423, 880,
          1815, 3719, 7582, 15397, 31171, 62952, 126891, 255379}
```

A closed form using the Fibonacci numbers:

```
In[69]:= Table[2^k - Fibonacci[k + 2], {k, 0, 18}]

Out[69]= {0, 0, 1, 3, 8, 19, 43, 94, 201, 423, 880,
          1815, 3719, 7582, 15397, 31171, 62952, 126891, 255379}
```

Problem 24

The problem is to find a recurrence realtion for the number of bit strings of length n which contain at least one occurences of the string 000.

Mathematica can select all such strings.

```
In[70]:= Select[Strings[{0, 1}, 4], MemberQ[Partition[#, 3, 1], {0, 0, 0}] &]

Out[70]= {{0, 0, 0, 0}, {0, 0, 0, 1}, {1, 0, 0, 0}}

In[71]:= Length[Select[Strings[{0, 1}, 4], MemberQ[Partition[#, 3, 1], {0, 0, 0}] &]]

Out[71]= 3

In[72]:= Select[Strings[{0, 1}, 5], MemberQ[Partition[#, 3, 1], {0, 0, 0}] &]

Out[72]= {{0, 0, 0, 0, 0}, {0, 0, 0, 0, 1}, {0, 0, 0, 1, 0}, {0, 0, 0, 1, 1},
          {0, 1, 0, 0, 0}, {1, 0, 0, 0, 0}, {1, 0, 0, 0, 1}, {1, 1, 0, 0, 0}}

In[73]:= Length[Select[Strings[{0, 1}, 5], MemberQ[Partition[#, 3, 1], {0, 0, 0}] &]]

Out[73]= 8

In[74]:= Table[
          Length[Select[Strings[{0, 1}, k], MemberQ[Partition[#, 3, 1], {0, 0, 0}] &]], {k, 1, 16}]

Out[74]= {0, 0, 1, 3, 8, 20, 47, 107, 238, 520, 1121, 2391, 5056, 10616, 22159, 46023}
```

Let $n \geq 3$. Denote by S the set of all bit strings of length n which contain at least one occurence of the string 000. The set S is a disjoint union of four sets: the set S_{000}, the set S_{100}, the set S_{10} and the set S_1. The set S_{000} is the set of all bit strings in S which end with 000, The set S_{100} is the set of all bit strings in S which end with 100, the set S_{10} is the set of all bit strings in S which end with 10, the set S_1 is the set of all bit strings in S which end with 1. The cardinality of S_{000} is 2^{n-3}, the cardinality of S_{100} is a_{n-3}, the cardinality of S_{10} is a_{n-2}, the cardinality of S_1 is a_{n-1}.

The corresponding recursion is

```
In[75]:= Clear[rs24]; rs24[0] = 0; rs24[1] = 0; rs24[2] = 0;
         rs24[n_] := rs24[n] = 2^(n-3) + rs24[n - 3] + rs24[n - 2] + rs24[n - 1]

In[77]:= Table[rs24[k], {k, 0, 19}]

Out[77]= {0, 0, 0, 1, 3, 8, 20, 47, 107, 238, 520, 1121,
          2391, 5056, 10616, 22159, 46023, 95182, 196132, 402873}
```

Problem 25

The problem is to find a recurrence realtion for the number of bit strings of length n which do not contain any occurences of the string 000.

Mathematica can select all such strings.

```
In[78]:= Select[Strings[{0, 1}, 4], Not[MemberQ[Partition[#, 3, 1], {0, 0, 0}]] &]

Out[78]= {{0, 0, 1, 0}, {0, 0, 1, 1}, {0, 1, 0, 0}, {0, 1, 0, 1},
          {0, 1, 1, 0}, {0, 1, 1, 1}, {1, 0, 0, 1}, {1, 0, 1, 0},
          {1, 0, 1, 1}, {1, 1, 0, 0}, {1, 1, 0, 1}, {1, 1, 1, 0}, {1, 1, 1, 1}}

In[79]:= Length[Select[Strings[{0, 1}, 4], Not[MemberQ[Partition[#, 3, 1], {0, 0, 0}]] &]]

Out[79]= 13

In[80]:= Table[Length[Select[Strings[{0, 1}, k], Not[MemberQ[Partition[#, 3, 1], {0, 0, 0}]] &]],
          {k, 0, 12}]

Out[80]= {1, 2, 4, 7, 13, 24, 44, 81, 149, 274, 504, 927, 1705}
```

Let $n \geq 3$. Denote by S the set of all bit strings of length n which do not contain any occurences of the string 000. The set S is a disjoint union of three sets: the set S_{100}, the set S_{10} and the set S_1. The set S_{100} is the set of all bit strings in S which end with 100, the set S_{10} is the set of all bit strings in S which end with 10, the set S_1 is the set of all bit strings in S which end with 1. The cardinality of S_{100} is a_{n-3}, the cardinality of S_{10} is a_{n-2}, the cardinality of S_1 is a_{n-1}.

The corresponding recursion is

```
In[81]:= Clear[rs25]; rs25[0] = 1; rs25[1] = 2; rs25[2] = 4;
         rs25[n_] := rs25[n] = rs25[n - 3] + rs25[n - 2] + rs25[n - 1]
```

The logic here is that rs[n-3] strings end with 100, rs[n-2] strings end with 10 and rs[n-1] strings end with 1.

```
In[83]:= Table[rs25[k], {k, 0, 22}]

Out[83]= {1, 2, 4, 7, 13, 24, 44, 81, 149, 274, 504, 927, 1705, 3136,
          5768, 10609, 19513, 35890, 66012, 121415, 223317, 410744, 755476}
```

Problem 26

The problem is to find a recurrence realtion for the number of bit strings of length n which contain at least one occurence of the string 01.

Easily we calculate $a_0 = 0,\ a_1 = 0,\ a_2 = 1,\ a_3 = 4$. Below is the *Mathematica* code confirming these calculations and calculating more values.

```
In[84]:= Select[Strings[{0, 1}, 4], MemberQ[Partition[#, 2, 1], {0, 1}] &]

Out[84]= {{0, 0, 0, 1}, {0, 0, 1, 0}, {0, 0, 1, 1}, {0, 1, 0, 0}, {0, 1, 0, 1},
          {0, 1, 1, 0}, {0, 1, 1, 1}, {1, 0, 0, 1}, {1, 0, 1, 0}, {1, 0, 1, 1}, {1, 1, 0, 1}}

In[85]:= Length[Select[Strings[{0, 1}, 4], MemberQ[Partition[#, 2, 1], {0, 1}] &]]

Out[85]= 11

In[86]:= Table[
          Length[Select[Strings[{0, 1}, k], MemberQ[Partition[#, 2, 1], {0, 1}] &]], {k, 1, 8}]

Out[86]= {0, 1, 4, 11, 26, 57, 120, 247}
```

Now reasoning which leads to the recursion.

Reasoning I.

Let $n \geq 2$. Denote by S the set of all bit strings of length n which contain at least one occurence of the string 01. The set S is a disjoint union of two sets: the set S_f and the set S_l. The set S_f is the set of all bit strings of length n in which the first occurence of 01 is at the position k where $k \leq n-2$. (Here we mean that the bit 0 in 01 is at the position k.) The set S_l is the set of all bit strings of length n in which the **first** occurence of 01 is at the position $n-1$, that is 0 is at the position $n-1$ and 1 is at the position n.

The cardinality of the set S_f is $2\,a_{n-1}$. To justify this claim, observe that each string in S_f is obtained by appending 0 or 1 at the end of a bit strings of length $n-1$ which contain at least one occurence of the string 01.

The cardinality of the set S_l is $n-1$. The justification for this claim is as follows. If the first occurence of 01 is at the last two bits, then the first $n-2$ positions do not include any 01 strings. There are $n-1$ such bit strings: all $n-2$ bits are 0, 1-s up to k, then zeros, $k = 1, \ldots, n-3$, all $n-2$ bits are 1.

The recursion corresponding to this disjoint union is

```
In[87]:= Clear[rs26a, n];
         rs26a[0] = 0; rs26a[n_] := rs26a[n] = 2 rs26a[n - 1] + n - 1

In[89]:= Table[rs26a[k], {k, 0, 18}]

Out[89]= {0, 0, 1, 4, 11, 26, 57, 120, 247, 502, 1013,
          2036, 4083, 8178, 16369, 32752, 65519, 131054, 262125}
```

We will present a different reasoning that leads to the same recursion.

Let $n \geq 2$. Denote by S the set of all bit strings of length n which contain at least one occurence of the string 01. The set S is a disjoint union of two sets: the set S_0 and the set S_1. The set S_0 is the set of all bit strings S which end with 0. The set S_1 is the set of all bit strings in S which end with 1.

The cardinality of the set S_0 is a_{n-1}. To justify this claim, observe that each string in S_0 is obtained by appending 0 at the end of a bit strings of length $n-1$ which contain at least one occurence of the string 01.

The cardinality of the set S_1 is a little more complicated. There are two kinds of bit strings in S_1. The first kind are bit strings which are obtained by appending 1 at the end of a bit strings of length $n-1$ which contain at least one occurence of the string 01. There are a_{n-1} bitstrings of this kind. The second kind are bit strings that end with 01 but do not contain the string 01 otherwise. There are $n-1$ strings like that. They are 00...01, 100..01, and so on to, 11...101; counting 0s they can have $n-1$ zeros, $n-2$ zeros, up to only 1 zero.

The recursion corresponding to this disjoint union is the same as one defined in rs26a.

Reasoning II.

Let $n \geq 2$. Denote by S the set of all bit strings of length n which contain at least one occurence of the string 01. The set S is a disjoint union of three sets the set S_0, the set S_{01} and the set S_{11}. The set S_0 is the set of all bit strings in S which end with 0. These bit strings are all obtained by appending 0 to bit strings of length $n-1$ which contain at least one occurence of the string 01. **There are** exactly a_{n-1} such strings. The set S_{01} is the set of all bit strings of length n which end by 01. **There are** exactly 2^{n-2} such bit strings. The set S_{11} is the set of all bit strings of length n which end by 11. These bit strings are all obtained by appending 1 to bit strings of length $n-1$ which contain at least one occurence of the string 01 and which end by 1. Since there are a_{n-2} bit strings of length $n-1$ which contain at least one occurence of the string 01 and which end by 0, **there are** exactly $a_{n-1} - a_{n-2}$ bit strings of length $n-1$ which contain at least one occurence of the string 01 and which end by 1.

The recursion corresponding to this disjoint union is

```
In[90]:= Clear[rs26b]; rs26b[0] = 0; rs26b[1] = 0;
         rs26b[n_] := rs26b[n] = rs26b[n - 1] + 2^(n-2) + rs26b[n - 1] - rs26b[n - 2]
```

```
In[92]:= Table[rs26b[k], {k, 0, 18}]

Out[92]= {0, 0, 1, 4, 11, 26, 57, 120, 247, 502, 1013,
          2036, 4083, 8178, 16369, 32752, 65519, 131054, 262125}
```

Closed form for the sequence

```
In[93]:= Table[2^n - (n + 1), {n, 0, 18}]

Out[93]= {0, 0, 1, 4, 11, 26, 57, 120, 247, 502, 1013,
          2036, 4083, 8178, 16369, 32752, 65519, 131054, 262125}
```

Problem 27

Count the number of ways to climb n stairs if we can take either 1 or 2 stairs at the time.

For example 3 stairs:

```
In[94]:= {{1, 2}, {2, 1}, {1, 1, 1}}

Out[94]= {{1, 2}, {2, 1}, {1, 1, 1}}
```

or 4 stairs

```
In[95]:= {{1, 1, 2}, {2, 2}, {1, 2, 1}, {2, 1, 1}, {1, 1, 1, 1}}

Out[95]= {{1, 1, 2}, {2, 2}, {1, 2, 1}, {2, 1, 1}, {1, 1, 1, 1}}

In[96]:= Select[Flatten[Table[Strings[{1, 2}, k], {k, 1, 5}], 1], (Total[#] == 5) &]

Out[96]= {{1, 2, 2}, {2, 1, 2}, {2, 2, 1}, {1, 1, 1, 2},
          {1, 1, 2, 1}, {1, 2, 1, 1}, {2, 1, 1, 1}, {1, 1, 1, 1, 1}}

In[97]:= Length[Select[Flatten[Table[Strings[{1, 2}, k], {k, 1, 5}], 1], (Total[#] == 5) &]]

Out[97]= 8

In[98]:= Table[Length[Select[Flatten[Table[Strings[{1, 2}, k], {k, 1, n}], 1], (Total[#] == n) &]],
          {n, 1, 12}]

Out[98]= {1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233}

In[99]:= Clear[rs27]; rs27[1] = 1; rs27[2] = 2;
         rs27[n_] := rs27[n] = rs27[n - 1] + rs27[n - 2]

In[101]:=
       Table[rs27[k], {k, 1, 12}]

Out[101]=
       {1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233}
```

Problem 28

```
In[102]:=
       Select[Flatten[Table[Strings[{1, 2, 3}, k], {k, 1, 4}], 1], (Total[#] == 4) &]

Out[102]=
       {{1, 3}, {2, 2}, {3, 1}, {1, 1, 2}, {1, 2, 1}, {2, 1, 1}, {1, 1, 1, 1}}

In[103]:=
       Select[Flatten[Table[Strings[{1, 2, 3}, k], {k, 1, 5}], 1], (Total[#] == 5) &]

Out[103]=
       {{2, 3}, {3, 2}, {1, 1, 3}, {1, 2, 2}, {1, 3, 1}, {2, 1, 2}, {2, 2, 1},
        {3, 1, 1}, {1, 1, 1, 2}, {1, 1, 2, 1}, {1, 2, 1, 1}, {2, 1, 1, 1}, {1, 1, 1, 1, 1}}

In[104]:=
       Length[Select[Flatten[Table[Strings[{1, 2, 3}, k], {k, 1, 5}], 1], (Total[#] == 5) &]]

Out[104]=
       13
```

In[105]:=
```
Table[
 Length[Select[Flatten[Table[Strings[{1, 2, 3}, k], {k, 1, n}], 1], (Total[#] == n) &]],
 {n, 1, 12}]
```

Out[105]=
```
{1, 2, 4, 7, 13, 24, 44, 81, 149, 274, 504, 927}
```

In[106]:=
```
Clear[rs28]; rs28[1] = 1; rs28[2] = 2; rs28[3] = 4;
rs28[n_] := rs28[n] = rs28[n - 1] + rs28[n - 2] + rs28[n - 3]
```

In[108]:=
```
Table[rs28[k], {k, 1, 12}]
```

Out[108]=
```
{1, 2, 4, 7, 13, 24, 44, 81, 149, 274, 504, 927}
```

29

a0 = 1; a1 = 3; a2 = 8;

In[109]:=
```
{{}, {0}, {1}, {2}}
```

Out[109]=
```
{{}, {0}, {1}, {2}}
```

In[110]:=
```
Length[{{1, 0}, {2, 0}, {0, 1}, {0, 2}, {1, 1}, {1, 2}, {2, 1}, {2, 2}}]
```

Out[110]=
```
8
```

Split the set of all ternary strings of length n with no consecutive 0s into disjoint subsets: beginning with 1, begining with 2 and beginning with 0. How many of each?

In[111]:=
```
Clear[ts]; ts[0] = 1; ts[1] = 3; ts[n_] := ts[n] = 2 * ts[n - 1] + 2 * ts[n - 2]
```

In[112]:=
```
Table[ts[k], {k, 1, 10}]
```

Out[112]=
```
{3, 8, 22, 60, 164, 448, 1224, 3344, 9136, 24960}
```